

算术运算符

五个算术 (arithmetic) 运算符, 按计算器使用它们的顺序排列。

运算符	含义	A 为 7,B 为 2
+	加	A + B 得 9
-	减	A - B 得 5
*	乘	A * B 得 14
/	除	A / B 得 3.5
^	幂	A ^ B 得 49

```
DECLARE A, B : INTEGER
A ← 7
B ← 2
OUTPUT "Sum: ", A + B
OUTPUT "Product: ", A * B
OUTPUT "Quotient: ", A / B
OUTPUT "Power: ", A ^ B
```

先算括号, 再算 ^, 然后是 * 和 /, 最后是 + 和 -—— 通常的优先级 (precedence)。只要读者需要停下来想一想, 就加括号。

```
DECLARE Total : REAL
Total ← (12 + 18) / 2
OUTPUT "Mean: ", Total
```

常见错误

- 以为 / 会给出整数。7 / 2 是 3.5, 不是 3; 要整数请用 DIV。
- 写成 A x B 或 A ÷ B。请用 * 和 /。

DIV 与 MOD

整数除法 (integer division) 回答两个不同的问题, 而 0478 把它们各写成一个函数 (function), 两个值放在括号里。

- DIV(a, b)——b 能整着进入 a 多少次, 即商 (quotient)。
- MOD(a, b)——剩下多少, 即余数 (remainder)。

```
OUTPUT DIV(10, 3)
OUTPUT MOD(10, 3)
OUTPUT DIV(17, 5)
OUTPUT MOD(17, 5)
```

两者都接受整数, 也都返回整数。

它们的用途

MOD 是”能否整除”的标准检验, 因而也是判断一个数是否为偶数的方法:

```
DECLARE Number : INTEGER
Number ← 12
IF MOD(Number, 2) = 0
  THEN
    OUTPUT Number, " is even"
  ELSE
    OUTPUT Number, " is odd"
ENDIF
```

DIV 把总量换算成整单位 —— 秒换分钟, 便士换英镑:

```
DECLARE Seconds : INTEGER
Seconds ← 200
OUTPUT DIV(Seconds, 60), " minutes and ", MOD(Seconds, 60), " seconds"
```

常见错误

- 10 MOD 3。那是 A-Level 9618 的写法;**0478 使用函数形式** MOD(10, 3)。
- 把两者弄混。DIV 给出整数部分;MOD 给出装不下的部分。

关系运算符与逻辑运算符

关系运算符 (relational operator) 比较两个值, 产生 TRUE 或 FALSE。

运算符	含义
=	等于
<>	不等于
>	大于
<	小于
>=	大于等于
<=	小于等于

```
DECLARE Mark : INTEGER
Mark ← 55
OUTPUT "Pass: ", Mark >= 50
OUTPUT "Perfect: ", Mark = 100
OUTPUT "Not zero: ", Mark <> 0
```

三个**逻辑运算符** (logic operator) 把条件连接起来:

- AND—— 两边都必须为 TRUE。
- OR—— 至少一边为 TRUE。
- NOT—— 把 TRUE 变成 FALSE, 反之亦然。

```
DECLARE Age : INTEGER
DECLARE HasTicket : BOOLEAN
Age ← 16
HasTicket ← TRUE
IF Age >= 15 AND HasTicket = TRUE
    THEN
        OUTPUT "Admitted"
ENDIF
IF NOT (Age > 18)
    THEN
        OUTPUT "Still a student rate"
ENDIF
```

常见错误

- 这里的不等于写作 <>, 而不是属于其他语言的 !=。
- IF Age >= 15 AND <= 18。AND 的每一边都需要完整的比较: Age >= 15 AND Age <= 18。
- 在答卷中称它们为比较运算符。考纲用语是**关系运算符**。